Curso de Curvas e Superfícies - Parte I

Wellington José Leite da Silva[1]

[1]Escola de Matemática Aplicada da FGV (EMAP), Brazil

Apresentação

Apresentamos aqui uma linha de aprendizado do curso de curvas e superfícies apresentando definições, teoremas, exemplos e etc. Separados em 6 seções sendo elas: Curvas regulares, Difeomorfismo e Reparametrização, Curvatura, Teorema Fundamental de Curvas Plana, Curvas Regulares no $\mathbb{R}^3$ e Teorema Fundamental das Curvas Espaciais.

Com intuído de auxiliar o aprendizado aos tópicos apresentados e fornecer uma forma de visualização computacional apresentamos exemplos com códigos em *SageMath* [The Sage Developers 2022]. Aqui seguimos o livro [de Lima 2016] como principal e o [do Carmo 2010] como complementar. Adicionando sempre que possível, exemplos de visualizações em *SageMath*. As implementações, códigos usados para as mesmas assim como o *Tex* deste documento se concentram no repositório curvas-superficies [1] que está disponível abertamente no github.

Todos os códigos apresentados nos exemplos podem ser facilmente generalizados para outros casos, é recomendável como forma de aprendizado rodar os códigos apresentados com outros exemplos de escolha do leitor.

1. Curvas regulares

Definição 1 *Uma **curva parametrizada** α em $\mathbb{R}^2$ é uma aplicação $\gamma : I \to \mathbb{R}^2$ sendo $I \subset \mathbb{R}$ aberto, da forma*

$$\alpha(t) = (x(t), y(t)),\ t \in I$$

onde x e y são funções diferenciáveis de t.

Exemplo 1.1 (Curva parametrizada diferenciável) *A curva*

$$\alpha(t) = (1/2\cos(3t) + 1/3\sin(t), 1/2\sin(3t) + 1/3\cos(t))$$

[1] https://github.com/wellington36/curvas-superficies

é um exemplo de curva parametrizada diferenciável e podemos visualizar no sagemath da seguinte forma

```
# Curve definition
curve_alpha(t) = (1/2 * cos(3 * t) + 1/3 * sin(t),
                  1/2 * sin(3 * t) + 1/3 * cos(t))

# Plot
parametric_plot(curve_alpha, (t,0, 2*pi), thickness=2)
```



Figure 1. Curva parametrizada

Definição 2 *O conjunto imagem de uma curva γ, $\gamma(I) \subset \mathbb{R}^n$ é dito o **traço** de γ.*

Definição 3 (Vetor tangente) *Seja $\gamma : I \to \mathbb{R}^n$ com $\gamma(t) = (\gamma_1(t), \ldots, \gamma_n(t))$ com $\gamma_i(y)$ diferenciáveis $\forall i, i = 1 \ldots n$, o vetor*

$$\gamma'(t) = (\gamma_1'(t), \ldots, \gamma_n'(t))$$

*é chamado **vetor tangente de** γ em t*

Exemplo 1.2 (Vetores tangentes) *Vamos mostrar os vetores tangentes a uma certa curva, aqui usamos o projeto SageManifolds* [2] *uma extensão do SageMath para geometria diferencial que está incluso no Sage, então podemos visualizar vetores tangentes a uma curva da seguinte forma*

```
# Define dimention and variables
M = Manifold(2, 'M')
X.<x,y> = M.chart()

# Define curve
c = M.curve([1/2 * cos(t) + 1/3 * sin(4 * t),
```

[2] https://sagemanifolds.obspm.fr/index.html

```
                 1/2 * sin(t) + 1/3 * cos(4 * t)], (t, 0, 2*pi))

# Vectors tangent
v = c.tangent_vector_field() ; v

# Plot
show(c.plot(thickness=2, aspect_ratio=1) +
      v.plot(chart=X, number_values=30, scale=.2))
```



Figure 2. Vetores tangentes

Definição 4 (Curvas regulares) *Seja $\gamma(t) : I \to \mathbb{R}^n$ uma curva parametrizada diferenciável. Diz-se que γ **é regular**, quando $\gamma'(t) \neq 0,\ \forall t \in I$.*

Definição 5 (Comprimento de arco) *O **comprimento de arco de** α, de $\alpha(a)$ até $\alpha(b)$ definido por $L_a^b(\alpha)$ é*

$$L_a^b(\alpha) = \int_a^b \|\alpha'(t)\| dt$$

Exemplo 1.3 (Comprimento de arco) *O comprimento de arco da curva $\alpha(t) = (2\cos(t), 2\sin(t))$ pode ser encontrada fazendo*

```
# Define curve
curve_alpha(s) = (2 * cos(s), 2 * sin(s))

# Calculate the derivative
x = get_vector_arguments(curve_alpha).pop()
curve_alpha_x = curve_alpha.derivative(x)

# Calcular comprimento de arco de 0 a t
t = var("t")
assume(t>0)
s = integrate(norm(curve_alpha_x), (x,0,t))
s = s.simplify_full()
```

```

# Plot equation
pretty_results((r"\int_0^t || C'(x) || dx", s))
```

$$\int_0^t ||C'(x)||dx = 2t$$

Definição 6 *Se $\gamma : (a,b) \to \mathbb{R}^n$ é uma c.p.*[3]*, sua **velocidade no ponto** $\gamma(t)$ é $\|\gamma'(t)\|$, e a curva é dita com **velocidade unitária** se $\|\gamma'(t)\| = 1,\ \forall t \in (a,b)$ e é parametrizada por comprimento de arco.*

Teorema 1 *Toda **curva regular** pode ser reparametrizada por **comprimento de arco**.*

2. Difeomorfismo e Reparametrizacao

Definição 7 (Difeomorfismo) *Dado os conjuntos abertos $U \subset \mathbb{R}^n$ e $V \subset \mathbb{R}^n$. Uma bijeção $f : U \to V$ é dita **difeomorfismo** quando f e f^{-1} são diferenciáveis.*

Definição 8 (Reparametrização) *A curva $\beta(s)$ é dita uma **reparametrização** de $\alpha(t) : I \subset \mathbb{R} \to \mathbb{R}^2$ regular quando dados $I_0 \subset \mathbb{R}$ e $\phi : I_0 \to I$ difeomorfismo. Temos $\beta(S) = \alpha(\phi(S)))$.*

Definição 9 *Seja $\alpha(t) : (a,b) \to \mathbb{R}^2$ r $\beta(S) : (c,d) \to \mathbb{R}^2$. Então*

- *$\beta(S)$ é uma reparametrização positiva de α se $\phi'(S) > 0,\ \forall S$*
- *$\beta(S)$ é uma reparametrização negativa de α se $\phi'(S) < 0,\ \forall S$*

Definição 10 *Qualquer reparametrização de uma c.p. regular é regular (i.e. difeomorfismos preservam regularidade).*

Exemplo 2.1 (Reparametrização por difeomorfismo) *Seja a seguinte curva (que vou chamar aqui de flor)*

$$\alpha(t) = (\cos(t) + \sin(4t), \sin(t) + \cos(4t))$$

podemos reparametrizar digamos pela função $\phi(t) = 2t + 1$ da seguinte forma, como de costume usando sagemath

[3] curva parametrizada

```
# Curve definition
alpha(t) = (cos(t) + sin(4 * t), sin(t) + cos(4 * t))

# Reparametrization
phi(x) = 2 * x + 1
beta(t) = alpha(phi(t))

# Plot
parametric_plot(alpha, (t,0, 2*pi), thickness=2)
parametric_plot(beta, (t,0, 2*pi), thickness=2, color='red')
```



Figure 3. Curva α



Figure 4. Curva α reparametrizada

Aqui ϕ é um difeomorfismo gerando uma reparametrização de α em particular o traço das curvas se mantém e β é uma reparametrização positiva

Proposição 1 *A função L (comprimento de arco) é um difeomorfismo.*

Definição 11 *Toda curva regular $\alpha : I \to \mathbb{R}^2$ admite reparametrização por comprimento de arco.*

Exemplo 2.2 (Reparametrização por comprimento de arco) *Usaremos aqui a mesma curva usada para calcular o comprimento de arco no caso o círculo de raio 2 $\alpha(t) = (2\cos(t), 2\sin(t))$, com ela podemos obter sua parametrização por comprimento de arco fazendo os seguintes passos*

```
# Curve definition
alpha(s) = (2 * cos(s), 2 * sin(s))

# Identify curve parameter
x = get_vector_arguments(alpha).pop()

# Calculate the derivatives
alpha_x = alpha.derivative(x)

# Calculate arc length from 0 to t
t = var("t")
assume(t>0)
comp_arco = integrate(norm(alpha_x), (x,0,t))

t = comp_arco.arguments()[0]

# Find t in terms of s
s = var("s")
param_comp_arco = solve(s == comp_arco, t)[0]

# Replace original parameter in curve
alpha_subs = alpha(t).subs(param_comp_arco)
alpha_subs = vector_simplify(alpha_subs)

# Reset function argument
alpha_param(s) = tuple(coord for coord in alpha_subs)

# New curve plot
print("Curva reparametrizada:")
pretty_results((r"\alpha(s)", alpha_param), use_colon=True)
```

Curva reparametrizada:

$$\alpha(s) : \quad s \mapsto \left(2\,\cos\left(\frac{1}{2}\,s\right), 2\,\sin\left(\frac{1}{2}\,s\right)\right)$$

A mesma sequência de comandos pode ser facilmente reproduzida em outras curvas.

3. Curvatura

Definição 12 (Função Ângulo) *Dada uma curva diferenciável $\gamma : I \to S^1$, onde S^1 é o círculo de $\mathbb{R}^2$ com centro na origem e raio 1, diz-se que $\theta : I \to \mathbb{R}$ é uma **função-ângulo** de γ, quando*

$$\gamma(s) = (\cos(\theta(s)), \sin(\theta(s)), \ \forall s \in I$$

Definição 13 (Curvatura) *Seja $\alpha : I \to \mathbb{R}$ unit-speed. Designando-se o vetor tangente de α em $s \in I$ por $T(s)$, podemos afirmar que a curva $T(s) = I \to S^1$ admite função ângulo*

$$T(s) = (\cos(\theta(s)), \sin(\theta(s)), \ \forall s \in I$$

*Daí a **curvatura** de α em $s \in I$ é definida por*

$$K(s) = \theta'(s) = det(\alpha'(s), \alpha''(s))$$

Exemplo 3.1 (Exemplo curvatura) *A curvatura da curva $\alpha(t) = (2\sin(t), 2\cos(t))$, pode ser calculada fazendo*

```
# Curve definition
curve(t) = (2 * cos(t), 2 * sin(t))

# Calculate the derivatives
curve_t = curve.derivative(t)
curve_tt = curve_t.derivative(t)

# Calculate J(c'(t))
curve_t_rotation(t) = (- curve_t[1], curve_t[0])

# Calculate the curvature
curvatura = curve_t_rotation.dot_product(curve_tt)/norm(curve_t)^3
curvatura = curvatura.simplify_full()

# Plot
pretty_results((r"K(s)", curvatura), use_colon=True)
```

$$K(s) : \ \frac{1}{2}$$

Em geral, a curvatura de $\alpha(t) = (r\sin(t), r\cos(t)),\ \forall r \in \mathbb{R} - \{0\}$ é $K_\alpha(t) = \frac{1}{r}$

4. Teorema Fundamental de Curvas Plana

Teorema 2 (Função-ângulo diferenciável) *Seja $\gamma : I \to S^1$ uma curva diferenciável. Então, γ admite uma função ângulo $\theta : I \to \mathbb{R}$, a qual é diferenciável. Além disso, toda função-ângulo de γ, a qual é diferenciável, difere de θ por uma constante.*

Corolário 2.1 *Seja $\alpha : I \subset \mathbb{R} \to \mathbb{R}$ e seja $\beta(s) = \alpha(\theta(s))$ a parametrização por comprimento de arco de α, a curvatura de α em $t \in I$ é $K_\alpha(t)$, e, por definição é a curvatura de β em $\theta^{-1}(t)$, isto é*

$$K_\alpha := K_\beta(\theta^{-1}(t))$$

Definição 14 (Diedro de Frenet) *Seja $\alpha : I \subset \mathbb{R} \to \mathbb{R}^2$ uma curva regular parametrizada por comprimento de arco. Dado $s \in I$, o vetor $N(s) = JT(s)$ é dito o vetor normal de α em $s \in I$. A base ortonormal de $\mathbb{R}^2$ formado por $T(s)$ e $N(s)$ é chamada **Dietro de Frenet** em s.*

Exemplo 4.1 (Diedro de Frenet) *Vamos encontrar aqui os vetores T e N do Diedro de Frenet da curva $\alpha(t) = (\sin(2t), \cos(t))$*

```
# Curve definition
alpha(t) = (sin(2 * t), cos(t))

# Diedro de Frenet
T(t) = alpha.derivative(t)/norm(alpha.derivative(t))

N(t) = (- T[1], T[0])

# Plot
pretty_results((r"T(t)", T))
pretty_results((r"N(t)", N))
```

$$T(t):\quad t \mapsto \left(\frac{2\,\cos\left(2\,t\right)}{\sqrt{4\,\cos\left(2\,t\right)^{2} + \sin\left(t\right)^{2}}}, -\frac{\sin\left(t\right)}{\sqrt{4\,\cos\left(2\,t\right)^{2} + \sin\left(t\right)^{2}}}\right)$$

$$N(t):\quad t \mapsto \left(\frac{\sin(t)}{\sqrt{4\cos(2t)^2+\sin(t)^2}}, \frac{2\cos(2t)}{\sqrt{4\cos(2t)^2+\sin(t)^2}}\right)$$

Definição 15 (Movimento Rígido) *$\Phi : \mathbb{R}^2 \to \mathbb{R}^2$ é dita **movimento rígido**, quando preserva distancia, isto é, para quaisquer $p, q \in \mathbb{R}^2$*

$$\|\Phi(p) - \Phi(q)\| = \|p - q\|$$

Teorema 3 *Seja $\Phi : A + p_0$ um movimento rígido direto de $\mathbb{R}^2$ e $\alpha : I \to \mathbb{R}^2$ uma curva regular parametrizada por comprimento de arco. Então, $\beta = \Phi \circ \alpha : I \to \mathbb{R}^2$ é uma curva regular de $\mathbb{R}^2$, parametrizada por comprimento de arco, tal que*

$$K_\alpha(s) = K_\beta(s)\ \forall s \in I$$

Teorema 4 (Teorema Fundamental da Teoria Local das Curvas Planas) *Sejam I um intervalo aberto da reta e $K : I \to \mathbb{R}$ uma função diferenciável.*

1. *Então existe uma curva diferenciavel $\alpha : I \to \mathbb{R}^2$, unit-speed, cuja função curvatura coincide com K.*
2. *Além disso, para toda $\beta : I \to \mathbb{R}^2$, unit-speed, que cumpre $K_\beta = K$, existe um movimento rígido $\Phi : \mathbb{R}^2 \to \mathbb{R}^2$ tal que $\alpha = \Phi \circ \beta$*

Exemplo 4.2 (Construção de curva a partir da curvatura) *Pelo teorema acima podemos construir uma curva a partir de uma função diferenciável (que será a curvatura) tomando, por exemplo, a função $K(t) = 1/t$ que seguimos da seguinte forma e obtemos uma curva cuja curvatura é K.*

```
# Define curvature and theta_0
curvatura(t) = 1/t
theta_0 = 0

# Identify the Curve Parameter
t = curvatura.arguments()[0]

# Build Angle Function
theta(t) = integrate(curvatura(t), t) + theta_0

# Build Curve From Angle Function
```

```
aux(t) = (cos(theta(t)), sin(theta(t)))

curva_ang(t) = integrate(aux(t), t)

curva_ang = vector_simplify(curva_ang)

# Export function with n-tuple
curva_construida = tuple(coord for coord in curva_ang)

# Plot
parametric_plot(curva_construida, (t,0, 2 * pi), thickness=2)
#pretty_results((r"C(t)", curva_construida))
```



Figure 5. Curva pela curvatura

5. Curvas Regulares no $\mathbb{R}^3$

Definição 16 (Curvas no $\mathbb{R}^3$) *As curvas diferenciáveis no $\mathbb{R}^3$, são definidas de forma análoga ao $\mathbb{R}^2$, isto é, uma **curva no** $\mathbb{R}^3$ é uma aplicação diferenciável de um intervalo I (aberto) em $\mathbb{R}^3$, da forma*

$$\alpha(t) = (x(t), y(t), z(t)),\ t \in I$$

*Onde x, y e z são diferenciáveis, e a curva é dita **regular** quando*

$$\alpha'(t) = (x'(t), y'(t), z'(t)) \neq (0, 0, 0),\ t \in I$$

Exemplo 5.1 (Curva no $\mathbb{R}^3$) *A curva*

$$\alpha(t) = (2t\sin(50t), 3t\cos(50t), t^2)$$

É um exemplo de curva regular no $\mathbb{R}^3$

```
# Curve definition
alpha(t) = (2 * t * sin(50 * t), 3 * t * cos(50 * t), t^2)

# Plot
parametric_plot(alpha, (t,0, 2*pi), thickness=2)
```



Figure 6. Curva no $\mathbb{R}^3$

Proposição 2 *De forma análoga vale para* $\mathbb{R}^3$ *que*

- *Comprimento de arco é invariável por reparametrização.*
- *Toda curva regular admite reparametrização unit-speed* ($\|\alpha'(t)\| = 1$).

Definição 17 (Curvatura no $\mathbb{R}^3$) *Dada uma curva* $\alpha : I \to \mathbb{R}$ *regular parametrizada por comprimento de arco, a* ***curvatura*** *de* α *em* $s \in I$ *é definida como*

$$K_\alpha(s) = \|\alpha''(s)\|$$

Definição 18 (2-regular) *Seja uma curva regular* $\alpha : I \to \mathbb{R}^3$ *unit-speed, e* $K_\alpha(s) > 0, \forall s$*, ou seja,* $\alpha''(s) \neq 0,\ \forall s$*. Então dizemos que* α *é* ***2-regular****.*

Definição 19 (Triedro de Frenet) *Para* α *2-regular, seja* $T(s) = \alpha'(s)$ *(vetor tangente),* $N(s) = \frac{\alpha''(s)}{\|\alpha''(s)\|}$ *(vetor normal) e* $B(s) = T(s) \times N(s)$ *(vetor binormal). Com estes* *vetores estabelecemos um referencial chamado **Triedro de Frenet** formado pelos vetores* $\{T(s), N(s), B(s)\}$*, onde,*

$$\begin{cases} B(s) = T(s) \times N(s) \\ N(s) = B(s) \times T(s) \\ T(s) = N(s) \times B(s) \end{cases}$$

Definição 20 (Curvatura e Torção) *Seja uma curva α 2-regular em $\mathbb{R}^3$ não necessariamente parametrizada por comprimento de arco, então a **curvatura e a torção** de α são definidas respectivamente como*

$$K_\alpha(t) = \frac{\|\alpha''(t) \times \alpha'(t)\|}{\|\alpha'(t)\|^3}$$

$$\mathcal{T}(t) = \frac{\langle(\alpha'(t) \times \alpha''(t)), \alpha'''(t)\rangle}{\|\alpha'(t) \times \alpha''(t)\|^2}$$

Exemplo 5.2 (Curvatura geral) *Podemos calcular a curvatura no $\mathbb{R}^3$ da seguinte forma*

```
# Curve
alpha(t) = (sin(2 * t), cos(2 * t), t)

# Calculate the derivatives
alpha_t = alpha.derivative(t)
alpha_tt = alpha_t.derivative(t)

# Apply General Curvature Formula
curvature = norm(alpha_t.cross_product(alpha_tt)) / norm(alpha_t)^3
curvature = curvature.simplify_full()

pretty_results((r"K(t)", curvature))
```

$$K(t) = \frac{4}{5}$$

Exemplo 5.3 (Torção) *Seguindo a mesma ideia do exemplo acima podemos calcular a torção*

```
# Curve
alpha(t) = (sin(2 * t), cos(2 * t), t)

# Calculate the derivatives
alpha_t = alpha.derivative(t)
```

```
alpha_tt = alpha_t.derivative(t)
alpha_ttt = alpha_tt.derivative(t)

# Get Vector Product of Velocity and Acceleration Vectors
prod_veloc_acel = alpha_t.cross_product(alpha_tt)

# Apply General Twist Formula
torcao = (prod_veloc_acel * alpha_ttt) / norm(prod_veloc_acel)^2
torcao = torcao.simplify_full()

pretty_results((r"\mathcal{T}(t)", torcao))
```

$$\mathcal{T}(t) = -\frac{2}{5}$$

Teorema 5 *Seja $\alpha : I \to \mathbb{R}^3$ uma curva 2-regular unit-speed, então*

$$\alpha \text{ é plana} \iff \mathcal{T}(s) \equiv 0, \ \forall s \in I$$

Teorema Fundamental das Curvas Espaciais

Teorema 6 (Teorema Fundamental da Teoria Local das Curvas Espaciais) *Sejam I um intervalo aberto, $K : I \to \mathbb{R}$ uma função positiva diferenciável e $\mathcal{T} : I \to \mathbb{R}$ uma função diferenciável*

1. *Então existe uma curva diferenciavel $\alpha : I \to \mathbb{R}^3$, unit-speed, tal que K e $\mathcal{T}$ concedem com a curvatura e torção de α respectivamente*
2. *Além disso, $\forall \beta : I \to \mathbb{R}^3$, unit-speed, que cumpre $K_\beta = K$ e $\mathcal{T}_\beta = \mathcal{T}$ existe um movimento rígido $\Phi : \mathbb{R}^3 \to \mathbb{R}^3$ tal que $\alpha(s) = \Phi(\beta(s))$*

References

de Lima, R. F. (2016). *INTRODUÇÃO À GEOMETRIA DIFERENCIAL*.

do Carmo, M. (2010). *Geometria diferencial de curvas e superfícies*. Textos Universitarios: Ciencias médicas. Sociedade Brasileira de Matemática.

The Sage Developers (2022). *SageMath, the Sage Mathematics Software System (Version 9.5)*. https://www.sagemath.org.